
Knihovna CFoxLib

1  ÚVOD
TXV 003 71.01

Knihovna CFoxLib je standardně dodávána jako součást programovacího prostředí
Mosaic. Knihovna obsahuje funkce a funkční  bloky podporující  možnosti  modulů v síti
CFox.

Pokud chceme funkce z knihovny CFoxLib použít v aplikačním programu PLC, je
třeba nejprve přidat tuto knihovnu do projektu. Knihovna je dodávaná jako součást in-
stalace prostředí Mosaic od verze 2012.3.

Knihovna CFoxLib

2  DATOVÉ TYPY
V knihovně CFoxLib jsou definovány následující datové typy:

Identifikátor Typ Význam

TStateC_RI_0401S ENUM Stavy bloku pro IR ovládání pro C-RI-0401S
(popis jednotlivých hodnot viz funkční blok 
fbC_RI_0401S)

TRCM2_1_Screen STRUCT Struktura popisující chování obrazovky zobrazené na 
modulu RCM2-1 (více viz funkční blok fbRCM2_1)

TC_EM_0401M_Com
mand

STRUCT Struktura pro uchování příkazů pro blok 
fbRead_C_EM_0401M a fbReadConf_C_EM_0401M

TC_EM_0300M_Com
mand

STRUCT Struktura pro uchování příkazů pro blok 
fbReadConf_C_EM_0300M

Dále jsou znovu definovány následující struktury modulů. Tyto struktury se využíva-
jí pro vstupně výstupní proměnné sloužící k předání dat jednotlivých modulů.

Identifikátor Typ Význam

TLIB_CRC_BTN STRUCT Struktura tlačítek C-RC-0002R

TLIB_CRC_DISP STRUCT Struktura displeje C-RC-0002R

TLIB_IRControl STRUCT Řídící byte IR vysílače C-RI-0401S

TLIB_IRin STRUCT Vstupní struktura C-RI-0401S

TLIB_IRout STRUCT Výstupní struktura C-RI-0401S

TLIB_IRStat STRUCT Stavový byte IR přijímače C-RI-0401S

TLIB_RCM_ICO STRUCT Struktura viditelnosti symbolů na displeji RCM2-1

TLIB_RCM2_OUT STRUCT Výstupní struktura RCM2-1

TLIB_CEM0401M_ARI STRUCT Vstupní struktura modulu C-EM-0401M

TLIB_CEM0401M_ARO STRUCT Výstupní struktura modulu C-EM-0401M

Knihovna CFoxLib

3  KONSTANTY
V knihovně CFoxLib nejsou definovány žádné konstanty

4  GLOBÁLNÍ PROMĚNNÉ
V knihovně CFoxLib nejsou definovány žádné globální proměnné.

5  FUNKCE

Knihovna CFoxLib obsahuje následující funkce:

Funkce Popis

DewPoint Vrací rosný bod [°C] na základě teploty v prostoru 
[°C] a relativní vlhkosti [%] (C-AQ-0004R)

RCM2_1_SyncVarReal Synchronizace editovatelné REAL proměnné na ob-
razovce RCM2-1, vrací TRUE pokud je obrazovka 
RCM2-1 nastavena pro zobrazení reálných čísel

RCM2_1_SyncVarTime Synchronizace editovatelné TIME proměnné na ob-
razovce RCM2-1, vrací TRUE pokud je obrazovka 
RCM2-1 nastavena pro zobrazení času

SolarEnergy Vrací hodnotu sluneční radiace [W/m^2], dle měřené-
ho napětí [mV] a teploty [°C] na čidle

C_EM_0401M_Init Inicializační funkce pro pole příkazů pro funkční blok 
fbRead_C_EM_0401M a fbReadConf_C_EM_0401M

C_EM_0401M_ReadLReal Příkaz pro čtení hodnoty LREAL pro funkční blok 
fbRead_C_EM_0401M a fbReadConf_C_EM_0401M

C_EM_0401M_ReadReal Příkaz pro čtení hodnoty REAL pro funkční blok 
fbRead_C_EM_0401M a fbReadConf_C_EM_0401M

C_EM_0401M_ReadReal2 Příkaz pro čtení dvou hodnot REAL pro funkční blok 
fbRead_C_EM_0401M a fbReadConf_C_EM_0401M

C_EM_0300M_Init Inicializační funkce pro pole příkazů pro funkční blok 
fbReadConf_C_EM_0300M

C_EM_0300M_ReadLReal Příkaz pro čtení hodnoty LREAL pro funkční blok 
fbReadConf_C_EM_0300M

C_EM_0300M_ReadReal Příkaz pro čtení hodnoty REAL pro funkční blok 
fbReadConf_C_EM_0300M

C_EM_0300M_ReadReal2 Příkaz pro čtení dvou hodnot REAL pro funkční blok 
fbReadConf_C_EM_0300M

Knihovna CFoxLib

5.1  Funkce DewPoint
Knihovna : CFoxLib

Funkce Dewpoint vrací rosný bod ve stupních Celsia na základě teploty v prostoru
a relativní  vlhkosti  v procentech.  Relativní vlhkost lze získat  například  pomocí  modulu
C-AQ-0004R.

Rosný bod je vypočítán dle následujícího vztahu a zaokrouhlen na celé stupně:

DewPoint=
5745,31

5745,31
Temp+ 273,15

−ln( Rh
100,0)

−273,15

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

Temp REAL Teplota v prostoru ve stupních Celsia

Rh REAL Relativní vlhkost v procentech

DewPoint

Návratová 
hodnota

REAL Rosný bod ve stupních Celsia

Příklad programu s voláním funkce DewPoint :

PROGRAM prgDewPointExample
  VAR
    DewPointTemp : REAL;
  END_VAR

  DewPointTemp := DewPoint(Temp := C_RC_0002R_IN.eTHERM,
                           Rh := C_AQ_0004R_IN.RH);

END_PROGRAM

Knihovna CFoxLib

5.2  Funkce RCM2_1_SyncVarReal
Knihovna : CFoxLib

Funkce RCM2_1_SyncVarReal je určena k použití  s blokem  fbRCM2_1. Funkce
umožňuje  synchronizaci  editovatelné  proměnné  typu  REAL  na  obrazovce  modulu
RCM2-1. Synchronizace je prováděna tak, aby bylo možné měnit proměnnou z více míst.
Toho je dosaženo tak, že se do proměnné zapisuje jen v okamžiku ukončení editace.

Funkce vrací hodnotu TRUE v případě, že je synchronizovaná proměnná na ob-
razovce nastavené pro zobrazení hodnot typu REAL.

Popis proměnných :
Proměnná Typ Význam

VAR_IN_OUT

Val REAL Synchronizovaná REAL proměnná

Screen TRCM2_1_Screen Struktura popisující obrazovku, na které se 
proměnná edituje

SyncVarReal

Návratová 
hodnota

BOOL TRUE  pokud je obrazovka RCM2-1 nastave-
na pro zobrazení reálných čísel

Příklad programu s voláním funkce  RCM2_1_SyncVarReal je součástí příkladu k
bloku fbRCM2_1.

Knihovna CFoxLib

5.3  Funkce RCM2_1_SyncVarTime
Knihovna : CFoxLib

Funkce  RCM2_1_SyncVarTime je určena k použití s blokem  fbRCM2_1. Funkce
umožňuje  synchronizaci  editovatelné  proměnné  typu  TIME  na  obrazovce  modulu
RCM2-1. Synchronizace je prováděna tak, aby bylo možné měnit proměnnou z více míst.
Toho je dosaženo tak, že se do proměnné zapisuje jen v okamžiku ukončení editace.

Funkce vrací hodnotu TRUE v případě, že je synchronizovaná proměnná na ob-
razovce nastavené pro zobrazení hodnot typu TIME.

Popis proměnných :
Proměnná Typ Význam

VAR_IN_OUT

Val TIME Synchronizovaná TIME proměnná

Screen TRCM2_1_Screen Struktura popisující obrazovku, na které se 
proměnná edituje

SyncVarTime

Návratová 
hodnota

BOOL TRUE  pokud je obrazovka RCM2-1 nastave-
na pro zobrazení času

Příklad programu s voláním funkce RCM2_1_SyncVarTime je součástí příkladu k
bloku fbRCM2_1.

Knihovna CFoxLib

5.4  Funkce SolarEnergy
Knihovna : CFoxLib

Vrací hodnotu sluneční radiace [W/m^2], dle měřeného napětí [mV] a teploty [°C]
na čidle.

Na vstup mV se přivede napětí měřené čidlem S-SI-01I v mV. Pro měření napětí
lze  využít  analogové  vstupy  následujících  modulů:  C-IT-0200I(-SI),  C-HM-0308M,
C(R)-HM-1113M nebo C(R)-HM-1121M s rozsahem 0 – 100 mV (respektive u C-IT-0200I
HI -100 ÷ 100mV).

Na vstupu  temp se očekává teplota měřená čidlem NTC 12k,  které je součástí
S-SI-01I.

Na vstup cal je pro přesné měření nutné zadat kalibrační konstantu konkrétního či-
dla S-SI-01I (konstanta je uvedena na štítku na vnitřní straně víčka a je potřeba ji opsat
před namontováním čidla)

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

mV REAL měřeného napětí [mV] na čidle

temp REAL teplota [°C] na čidle

cal REAL kalibrační konstanta čidla solární radiace S-SI-01I
mV při 1000W/m2 a 25°C

SolarEnergy

Návratová hodnota REAL Vrací hodnotu sluneční radiace [W/m^2], dle měřené-
ho napětí [mV] a teploty [°C] na čidle 

Knihovna CFoxLib

5.5  Funkce C_EM_0401M_Init
Knihovna : CFoxLib

Inicializační  funkce  pro  pole  příkazů  pro  funkční  blok  fbRead_C_EM_0401M a
fbReadConf_C_EM_0401M. Nutné zavolat jako první před voláním jednotlivých příkazů;
využije první položku pole, zbylé položky je možné použít pro vlastní příkazy.

Funkce vrací maximální počet příkazů na základě bytové velikosti pole předané na
vstupu size.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

size UDINT velikost pole příkazů získaní pomocí SIZEOF

VAR_IN_OUT

C_EM_0401M_
Command

TC_EM_0401M_
Command

první člen pole příkazů

C_EM_0401M_Init

Návratová 
hodnota

UINT maximální počet příkazů

Příklad programu viz fbRead_C_EM_0401M nebo fbReadConf_C_EM_0401M.

Knihovna CFoxLib

5.6  Funkce C_EM_0401M_ReadLReal
Knihovna : CFoxLib

Funkce vloží příkaz pro čtení jedné proměnné typu LREAL do pole příkazů pro
funkční blok fbRead_C_EM_0401M a fbReadConf_C_EM_0401M. 

Mapování proměnných do registrů je popsáno v TXV 004 13 – Periferní moduly na
sběrnici CIB.

Funkce vrací status čtené hodnoty.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

reg UINT čtený registr

VAR_IN_OUT

lreal1 LREAL cílová proměnná

C_EM_0401M_
Command

TC_EM_0401M_
Command

první člen pole příkazů

C_EM_0401M_ReadLReal

Návratová 
hodnota

USINT status čtené hodnoty
1 – nově založený příkaz
2 – čtená hodnota je platná
252 – vypršel čas na odpověď
253 – čtená hodnota není platná
254 – neplatné číslo registru
255 – v poli příkazů není dostatek místa

Příklad programu viz fbRead_C_EM_0401M nebo fbReadConf_C_EM_0401M.

Knihovna CFoxLib

5.7  Funkce C_EM_0401M_ReadReal
Knihovna : CFoxLib

Funkce vloží  příkaz  pro  čtení  jedné proměnné  typu REAL do pole  příkazů pro
funkční blok fbRead_C_EM_0401M a fbReadConf_C_EM_0401M.

Mapování proměnných do registrů je popsáno v TXV 004 13 – Periferní moduly na
sběrnici CIB.

Funkce vrací status čtené hodnoty.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

reg UINT čtený registr

VAR_IN_OUT

real1 REAL cílová proměnná

C_EM_0401M_
Command

TC_EM_0401M_
Command

první člen pole příkazů

C_EM_0401M_ReadReal

Návratová 
hodnota

USINT status čtené hodnoty
1 – nově založený příkaz
2 – čtená hodnota je platná
252 – vypršel čas na odpověď
253 – čtená hodnota není platná
254 – neplatné číslo registru
255 – v poli příkazů není dostatek místa

Příklad programu viz fbRead_C_EM_0401M nebo fbReadConf_C_EM_0401M.

Knihovna CFoxLib

5.8  Funkce C_EM_0401M_ReadReal2
Knihovna : CFoxLib

Funkce vloží příkaz pro čtení dvou proměnných typu REAL do pole příkazů pro
funkční blok fbRead_C_EM_0401M a fbReadConf_C_EM_0401M.

Mapování proměnných do registrů je popsáno v TXV 004 13 – Periferní moduly na
sběrnici CIB.

Funkce vrací status čtené hodnoty.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

reg UINT čtený registr

VAR_IN_OUT

real1 REAL cílová proměnná pro první REAL

real2 REAL cílová proměnná pro druhý REAL

C_EM_0401M_
Command

TC_EM_0401M_
Command

první člen pole příkazů

C_EM_0401M_ReadReal2

Návratová 
hodnota

USINT status čtených hodnot
1 – nově založený příkaz
2 – čtené hodnoty jsou platné
252 – vypršel čas na odpověď
253 – čtené hodnota nejsou platné
254 – neplatné číslo registru
255 – v poli příkazů není dostatek místa

Příklad programu viz fbRead_C_EM_0401M nebo fbReadConf_C_EM_0401M.

Knihovna CFoxLib

5.9  Funkce C_EM_0300M_Init
Knihovna : CFoxLib

Inicializační funkce pro pole příkazů pro funkční blok fbReadConf_C_EM_0300M.
Nutné zavolat jako první před voláním jednotlivých příkazů; využije první položku pole,
zbylé položky je možné použít pro vlastní příkazy.

Funkce vrací maximální počet příkazů na základě bytové velikosti pole předané na
vstupu size.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

size UDINT velikost pole příkazů získaní pomocí SIZEOF

VAR_IN_OUT

C_EM_0300M_
Command

TC_EM_0300M_
Command

první člen pole příkazů

C_EM_0300M_Init

Návratová 
hodnota

UINT maximální počet příkazů

Příklad programu viz fbReadConf_C_EM_0300M.

Knihovna CFoxLib

5.10  Funkce C_EM_0300M_ReadLReal
Knihovna : CFoxLib

Funkce vloží příkaz pro čtení jedné proměnné typu LREAL do pole příkazů pro
funkční blok fbReadConf_C_EM_0300M. 

Mapování proměnných do registrů je popsáno v TXV 004 13 – Periferní moduly na
sběrnici CIB.

Funkce vrací status čtené hodnoty.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

reg UINT čtený registr

VAR_IN_OUT

lreal1 LREAL cílová proměnná

C_EM_0300M_
Command

TC_EM_0300M_
Command

první člen pole příkazů

C_EM_0300M_ReadLReal

Návratová 
hodnota

USINT status čtené hodnoty
1 – nově založený příkaz
2 – čtená hodnota je platná
252 – vypršel čas na odpověď
253 – čtená hodnota není platná
254 – neplatné číslo registru
255 – v poli příkazů není dostatek místa

Příklad programu viz fbReadConf_C_EM_0300M.

Knihovna CFoxLib

5.11  Funkce C_EM_0300M_ReadReal
Knihovna : CFoxLib

Funkce vloží  příkaz  pro  čtení  jedné proměnné  typu REAL do pole  příkazů pro
funkční blok fbReadConf_C_EM_0300M.

Mapování proměnných do registrů je popsáno v TXV 004 13 – Periferní moduly na
sběrnici CIB.

Funkce vrací status čtené hodnoty.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

reg UINT čtený registr

VAR_IN_OUT

real1 REAL cílová proměnná

C_EM_0300M_
Command

TC_EM_0300M_
Command

první člen pole příkazů

C_EM_0300M_ReadReal

Návratová 
hodnota

USINT status čtené hodnoty
1 – nově založený příkaz
2 – čtená hodnota je platná
252 – vypršel čas na odpověď
253 – čtená hodnota není platná
254 – neplatné číslo registru
255 – v poli příkazů není dostatek místa

Příklad programu viz fbReadConf_C_EM_0300M.

Knihovna CFoxLib

5.12  Funkce C_EM_0300M_ReadReal2
Knihovna : CFoxLib

Funkce vloží příkaz pro čtení dvou proměnných typu REAL do pole příkazů pro
funkční blok fbReadConf_C_EM_0300M.

Mapování proměnných do registrů je popsáno v TXV 004 13 – Periferní moduly na
sběrnici CIB.

Funkce vrací status čtené hodnoty.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

reg UINT čtený registr

VAR_IN_OUT

real1 REAL cílová proměnná pro první REAL

real2 REAL cílová proměnná pro druhý REAL

C_EM_0300M_
Command

TC_EM_0300M_
Command

první člen pole příkazů

C_EM_0300M_ReadReal2

Návratová 
hodnota

USINT status čtených hodnot
1 – nově založený příkaz
2 – čtené hodnoty jsou platné
252 – vypršel čas na odpověď
253 – čtené hodnota nejsou platné
254 – neplatné číslo registru
255 – v poli příkazů není dostatek místa

Příklad programu viz fbReadConf_C_EM_0300M.

Knihovna CFoxLib

6  FUNKČNÍ BLOKY
V knihovně CFoxLib jsou definovány následující funkční bloky:

Funkční blok Popis

fbC_RC_0002R Zobrazení teploty, editace žádané teploty a přepínání 
režimu pro modul C-RC-0002R

fbC_RC_0002R_2 Zobrazení teploty, editace dvou žádaných teplot a 
přepínání režimu pro modul C-RC-0002R

fbC_RI_0401S Posílání, příjem a učení IR kódu pro C-RI-0401S

fbRCM2_1 Jednoduché řízení RCM2-1 pomocí pole definicí ob-
razovek

fbSoilMoisture Měření vlhkosti půdy pro C-HM-0308M, C-HM-1113 a 
C-HM-1121

fbRead_C_EM_0401M Blok pro čtení dodatečných parametrů z registrů mo-
dulu C-EM-0401M

fbReadConf_C_EM_0401M Blok pro čtení dodatečných parametrů z registrů mo-
dulu a konfiguraci proudových traf C-EM-0401M

fbReadConf_C_EM_0300M Blok pro čtení dodatečných parametrů z registrů mo-
dulu a konfiguraci proudových traf C-EM-0300M

Knihovna CFoxLib

6.1  Funkční blok  fbC_RC_0002R
Knihovna : CFoxLib

Funkční blok fbC_RC_0002R slouží pro obsluhu modulu C-RC-0002R. Tento blok
implementuje tyto funkce:

• zobrazení hodnoty (typicky teplota v prostoru)
• editace teploty (typicky žádaná teplota)
• přepínání režimu (typicky útlum / komfort)

Zobrazená čísla jsou v rozsahu -9,9 až 99,9 s jedním desetinným místem a -99 až
999 bez desetinných míst. Standardně je zobrazena hodnota přivedená na vstup Temp.
Po zmačknutí  jednoho z tlačítek modulu je zobrazena editovatelná hodnota ze vstupu
TempReq. Tuto hodnotu lze měnit pomocí tlačítek UP a DOWN. Změna hodnoty má v
rozsahu -9,5 až 99,5 krok 0,5, pro hodnoty nižší než -9,5 a vyšší než 99,5 je editační krok
1. První stisk tlačítka hodnotu pouze zobrazí, až následující stisk nebo podržení tlačítka
déle jak půl sekundy způsobí změnu hodnoty.

Editovaná hodnota je po posledním stisku tlačítka zobrazována pět sekund. Po-
slední dvě sekundy hodnota bliká.

Tlačítko MODE alternuje proměnnou na vstupu  Mode. Stav proměnné na vstupu
Mode je indikován na modulu LED diodou se symbolem měsíčku. LED dioda je rozsvíce-
na pokud je hodnota proměnné TRUE.

Zápis do proměnných na vstupech TempReq a  Mode probíhá jen při změně, což
umožňuje ovládaní z více míst.

Spárování funkčního bloku s konkrétním modulem je realizováno přes vstupy Btn a
Disp. Na vstupu Btn se předává struktura BTN obsahující stavy tlačítek a na vstupu Disp
struktura  DISP řídící  zobrazení. Při volání v jazyku ST je potřeba při předávání těchto
struktur použít funkci void, jinak bude hlášen nesoulad datového typu.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

Temp REAL Zobrazovaná teplota (typicky teplota v prosto-
ru)

VAR_IN_OUT

Btn TLIB_CRC_BTN Struktura modulu TCIB_CRC_BTN

Disp TLIB_CRC_DISP Struktura modulu TCIB_CRC_DISP

TempReq REAL Zobrazená hodnota editovaná tlačítky UP a 
DOWN (typicky žádaná teplota)

Mode BOOL Režim

VAR_OUTPUT

Edit BOOL TRUE znamená, že je zobrazena editovatelná 
hodnota

Err BOOL TRUE znamená, že zobrazované číslo je mimo
rozsah

Příklad programu s funkčním blokem fbC_RC_0002R. Proměnné C_RC_0002R_IN
a  C_RC_0002R_OUT jsou  uživatelské  názvy  vstupní  a  výstupní  zóny  modulu
C-RC-0002R.

VAR_GLOBAL RETAIN
  ReqTemp : REAL := 22.0;
  Mode    : BOOL;
END_VAR

PROGRAM prgC_RC_Example
  VAR
    C_RC_1 : fbC_RC_0002R;
  END_VAR

  C_RC_1(Temp := C_RC_0002R_IN.iTHERM,
         Btn := void(C_RC_0002R_IN.BTN),
         Disp := void(C_RC_0002R_OUT.DISP),
         TempReq := ReqTemp,
         Mode := Mode);

END_PROGRAM

6.2  Funkční blok fbC_RC_0002R_2
Knihovna : CFoxLib

Funkční blok  fbC_RC_0002R_2 je variantou bloku  fbC_RC_0002R.  Blok se liší
možností  editovat  pro  každý  režim  vlastní  žádanou  teplotu.  Hodnota  na  vstupu
TempReq1 je  editována  v  případě,  když  je  proměnná  Mode nastavena  na  FALSE.
Hodnota TempReq2 pokud je Mode TRUE.

Typické použití může být editace žádané teploty pro běžný provoz a útlum.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

Temp REAL Zobrazovaná teplota (typicky teplota v prosto-
ru)

VAR_IN_OUT

Btn TLIB_CRC_BTN Struktura modulu TCIB_CRC_BTN

Disp TLIB_CRC_DISP Struktura modulu TCIB_CRC_DISP

TempReq1 REAL Zobrazená hodnota editovaná tlačítky UP a 
DOWN v případě Mode rovno FALSE
(typicky žádaná teplota pro běžný provoz)

TempReq2 REAL Zobrazená hodnota editovaná tlačítky UP a 
DOWN v případě Mode rovno TRUE
(typicky žádaná teplota pro útlum)

Mode BOOL Režim

VAR_OUTPUT

Edit BOOL TRUE znamená, že je zobrazena editovatelná
hodnota

Err BOOL TRUE znamená, že zobrazované číslo je 
mimo rozsah

Příklad programu s funkčním blokem fbC_RC_0002R_2:

V příkladu jsou použity  bloky z knihovny RegoLib.  Příklad realizuje  jednoduché
řízení  topení  s  týdenním programem  Prog2.  Výstup týdenního  programu je  negovaně
promítán do režimu indikovaného modulem C-RC-0002R.

Algoritmus umožňuje změnit režim pomocí tlačítka MODE. Tento vnucený režim
trvá buď jednu hodinu (časovač  ManTimer) nebo do první změny režimu dle týdenního
programu.

Topení je ovládáno blokem Heating realizující spínaní s hysterezí půl stupně.
Proměnné C_RC_0002R_IN a C_RC_0002R_OUT jsou uživatelské názvy vstupní

a výstupní zóny modulu C-RC-0002R.

VAR_GLOBAL RETAIN
  WeekProg  : _TimeProg2_Week_ := (
    Mon := ( T_ON1 := T#6h,  T_OFF1 := T#7h,
             T_ON2 := T#16h, T_OFF2 := T#22h),
    Tue := ( T_ON1 := T#6h,  T_OFF1 := T#7h,
             T_ON2 := T#16h, T_OFF2 := T#22h),
    Wed := ( T_ON1 := T#6h,  T_OFF1 := T#7h,
             T_ON2 := T#16h, T_OFF2 := T#22h),
    Thu := ( T_ON1 := T#6h,  T_OFF1 := T#7h,
             T_ON2 := T#16h, T_OFF2 := T#22h),
    Fri := ( T_ON1 := T#6h,  T_OFF1 := T#7h,
             T_ON2 := T#16h, T_OFF2 := T#22h),
    Sat := ( T_ON1 := T#8h,  T_OFF1 := T#22h,
             T_ON2 := T#0s,  T_OFF2 := T#0s),
    Sun := ( T_ON1 := T#8h,  T_OFF1 := T#22h,
             T_ON2 := T#0s,  T_OFF2 := T#0s));

  TempReqLo : REAL := 21.0;
  TempReqHi : REAL := 23.0;
END_VAR

PROGRAM prgC_RC_2_Example
  VAR
    Prog2      : TProg2;
    Heating    : Hyst1;
    C_RC_2     : fbC_RC_0002R_2;
    lastMode   : BOOL := 1;
    crcMode    : BOOL;
    ManTimer   : TON := ( PT := T#1h );
  END_VAR

  Prog2(TPg := WeekProg);

  IF Prog2.Out <> lastMode THEN
    lastMode := Prog2.Out;
    crcMode  := NOT lastMode;
  END_IF;

  ManTimer(IN := crcMode = lastMode);

  IF ManTimer.Q THEN
    crcMode  := NOT lastMode;
  END_IF;

  C_RC_2(Temp := C_RC_0002R_IN.iTHERM,
         Btn := void(C_RC_0002R_IN.BTN),
         Disp := void(C_RC_0002R_OUT.DISP),
         TempReq1 := TempReqHi,
         TempReq2 := TempReqLo,
         Mode := crcMode);

  IF crcMode THEN
    Heating.IN := TempReqLo;
  ELSE
    Heating.IN := TempReqHi;
  END_IF;

  Heating(PV := C_RC_0002R_IN.iTHERM, HV := 0.5);

END_PROGRAM

6.3  Funkční blok fbC_RI_0401S
Knihovna : CFoxLib

Funkční blok fbC_RI_0401S slouží k posílání, příjem a učení IR kódů pomocí mo-
dulu C-RI-0401S.

Spárování funkčního bloku s konkrétním modulem je realizováno přes vstupy IRin
a  IRout. Na vstupu  IRin se předává vstupní struktura  IRin a na vstupu  IRout výstupní
struktura modulu IRout. Při volání v jazyku ST je potřeba při předávání těchto struktur po-
užít funkci void, jinak bude hlášen nesoulad datového typu.

Příjem:
Ve výchozím stavu blok převádí modul do režimu příjmu. Připravenost k příjmu IR

kódu je  indikována hodnotou FALSE na výstupu  Busy.  Během příjmu kódu je  výstup
Busy nastaven na TRUE. Úspěšné přijmutí kódu je indikováno nastavení výstupu Rece-
iveDone na jeden cyklus na TRUE. Pro úspěšný příjem kódu je nutné, aby se tento kód
modul nejdříve naučil (viz níže). Příslušný index na který byl přiřazen kódu při učení je při
příjmu indikován na výstupu CodeRecv.

Učení:
Režim učení se iniciuje nastavením vstupu Learn na TRUE. Úspěšný přechod do

režimu  učení  je  indikováno  nastavením  výstupu  Busy na  TRUE  a  State na
sC_RI_0401S_Learning. Při přechodu do režimu učení musí být na vstupu  CodeLearn
nenulové číslo menší než 65536 určující index kódu, který bude použit při příjmu a vysí-
lání. Úspěšné naučení nového kódu je indikováno nastavením výstupu LearnDone na je-
den cyklus na TRUE.

Vysílání:
Naučené IR kódy lze pomocí modulu vyslat. Vysílání se zahájí nastavením vstupu

Transmit na TRUE. Při zahájení vysílání musí být na vstupu CodeTrans index kódu zvo-
lený při učení. Některá zařízení vyžadují vícenásobné vyslání kódu, aby byl přijat. Počet

opakování vysílaného kódu lze nastavit na vstupu Repeats. Pokud je Repeats roven nule
IR kód je vyslán jednou, pokud je roven jedné je vyslán dvakrát, atd.

Během vysílání je výstup Busy nastaven na TRUE a výstup State nabývá hodnot
sC_RI_0401S_Trasmiting a sC_RI_0401S_GettingReady.  Úspěšné vyslání je indikováno
nastavením výstupu TransmitDone na TRUE po dobu jednoho cyklu.

Mazání:
Naučené IR kódy lze z modulu vymazat. Mazání je možné buď selektivně, zadáním

indexu k vymazání na vstupu  CodeClear a nastavením  Clear na TRUE, nebo globálně
nastavením ClearAll na TRUE. Během mazání je výstup Busy nastaven na TRUE a vý-
stup  State nabývá  hodnot  sC_RI_0401S_Clearing a  sC_RI_0401S_GettingReady.
Úspěšné  vymazání  je  indikováno  nastavením  výstupu  ClearDone na  TRUE  po  dobu
jednoho cyklu.

Pokud dojde během práce s IR kódy k chybě je nastaven výstup Err na TRUE a vý-
stup ErrId na nenulovou hodnotu. Význam hodnot, kterých může výstup ErrId nabýt je po-
psán v následující tabulce:

ErrId Význam

1 Chyba při příjmu. Přijatý kód neodpovídá žádnému naučenému kódu.

2
Chyba učení. Nebyl nastaven platný index pro kód k naučení nebo modul 
nebyl schopen zpracovat IR kód, který se měl naučit.

3
Chyba mazání. Nebylo zadán platný index pro mazání nebo se nepodařilo
kód smazat.

4
Požadovaný režim není možné nastavit. Tato chyba může nastat v přípa-
dě, že kromě aktuálního bloku zasahuje do vstupní struktury modulu 
uživatel nebo jiný blok.

5 Během 30 sekund od započetí učení nebyl přijat platný IR kód.

Během provozu blok prochází následujícími stavy:

State Význam

0 sC_RI_0401S_Receiving
Výchozí stav, modul je připraven k příjmu a může 
být převeden do jakéhokoli jiného režimu.

1 sC_RI_0401S_Learning
Učící mód modul čeká po dobu 30 sekund na nový 
IR kód k naučení.

2 sC_RI_0401S_Trasmiting Modul vysílá IR kód.

3 sC_RI_0401S_Clearing Modul maže jeden nebo více naučených IR kódů.

4 sC_RI_0401S_Error Nastala chyba.

5
sC_RI_0401S_Get-
tingReady

Přechod do výchozího stavu.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

Transmit BOOL R_EDGE Vyšle kód daný indexem CodeTrans

Learn BOOL R_EDGE Zahájí mód učení pro index CodeLearn

Cancel BOOL R_EDGE Ukončí jakýkoli probíhající mód

Clear BOOL R_EDGE Vymaže naučený kód na indexu CodeClear

ClearAll BOOL R_EDGE Vymaže všechny naučené kódy

Repeats UINT Udává počet opakování kódu během vysílání

CodeTrans UINT Index kódu k vyslání

CodeLearn UINT Index kódu k naučení

CodeClear UINT Index kódu k vymazání

VAR_IN_OUT

IRin TLIB_IRin Struktura IRin modulu C-RI-0401S

IRout TLIB_IRout Struktura IRout modulu C-RI-0401S

VAR_OUTPUT

ReceiveDone BOOL Byl přijat nový kód

LearnDone BOOL Modul se naučil nový kód

TransmitDone BOOL Kód byl vyslán

ClearDone BOOL Mazání bylo dokončeno

Busy BOOL Modul je zaneprázdněn

Err BOOL Nastala chyba

ErrId UINT Identifikační kód chyby

State TStateC_RI_0401S Stav obsluhy modulu

CodeRecv UINT Index přijatého kódu

Příklad programu s funkčním blokem fbC_RI_0401S. Příklad demonstruje možnost
učení, vysílání a přijímání 32 uživatelsky pojmenovaných kódů s vazbou na ovládání přes
webové  rozhraní.  Uživatelská  jména  jsou  uložena  v  zálohované  proměnné  IrCom-
mandNames.  Ovládání  je  realizováno  přes  proměnnou  IrCommand,  která  obsahuje
všechny proměnné potřebné pro ovládání přes webové rozhraní.

Význam položek struktury proměnné IrCommand.

Položka Význam

Index Index aktuálního IR kódu

Next Přechod na o jedna vyšší index

Prev Přechod na o jedna nižší index

Learn Zahájení učení pro aktuální index

Test Testování aktuálního indexu, očekává se příjem kódu naučeného na 
vybraný index. Je nastaveno automaticky po úspěšném ukončení 
učení.

TestOk Příznak úspěšně dokončeného testu

Auto Příznak automatického učení. Po naučení a otestování kódu se auto-
maticky přejde na následující index

Send Vyslání IR kódu aktuálního indexu

Clear Vymazání naučeného IR kódu aktuálního indexu

ClearAll Vymazání všech naučených IR kódů

Transmit Řetězec 32 bitů ovládající vysílání kódů. Nastavením bitu 0 na hodno-
tu TRUE se vyšle kód s indexem 1, nastavením bitu 1 kód s indexem 
2 atd. Pro ovládání bitu lze použít tečkový zápis (např. IrCom-
mand.Transmit.5)

Text Popis aktuálního indexu

TextLast Poslední programem nastavený text aktuálního indexu. Neshoda s 
položkou Text je detekcí modifikace z webového rozraní. Obsah 
položky Text  je při promítnut do proměnné  IrCommandNames.

TextRecv Textový popis indexu posledního přijatého IR kódu.

Položky  Next, Prev, Learn, Test, Auto, Send, Clear, ClearAll,  Transmit.0, Trans-
mit.1,   …   Transmit.31 proměnné  IrCommand jsou  navržené  pro  použití  jako  řídící
proměnné dvoustavových obrázků ve funkci ovládacích tlačítek. Položka  TestOk slouží
jako indikace úspěšného textu. Položka  Text může být zobrazena v editovatelném poli,
které umožní uživatelskou změnu jména IR kódu.

TYPE
  TIrCommand : STRUCT
    Index    : UINT;
    Next     : BOOL;
    Prev     : BOOL;
    Learn    : BOOL;
    Test     : BOOL;
    TestOk   : BOOL;
    Auto     : BOOL;
    Send     : BOOL;
    Clear    : BOOL;
    ClearAll : BOOL;
    Transmit : DWORD;
    Text     : STRING[24];
    TextLast : STRING[24];
    TextRecv : STRING[24];
  END_STRUCT;
END_TYPE

VAR_GLOBAL CONSTANT
  IrCommandMax  : UINT := 31;
END_VAR

VAR_GLOBAL RETAIN
  IrCommandNames : ARRAY [0..IrCommandMax] OF STRING [24]:= [
    'Power',      'Mute',       'Volume +',   'Volume -',
    'Return',     'Start',      'Cancel',     'Up',
    'Down',       'Left',       'Right',      'Ok',
    'Stop',       'Play/Pause', 'Previous',   'Next',
    'Command 17', 'Command 18', 'Command 19', 'Command 20',
    'Command 21', 'Command 22', 'Command 23', 'Command 24',
    'Command 25', 'Command 26', 'Command 27', 'Command 28',
    'Command 29', 'Command 30', 'Command 31', 'Command 32'
    ];
END_VAR

VAR_GLOBAL
  IrCommand : TIrCommand;
END_VAR

FUNCTION DwordBitIndex : USINT
  VAR_INPUT
    In : DWORD;
  END_VAR
  VAR
    i : USINT;
  END_VAR

  WHILE In > 0 DO
    In := SHR(IN := In, N := 1);
    i := i + 1;
  END_WHILE;
  DwordBitIndex := i;

END_FUNCTION

PROGRAM prgC_RI_Example
  VAR
    IrControl         : fbC_RI_0401S;
    IrCommandLearnEnd : F_TRIG;
    AutoTimeOk        : TON;
  END_VAR

  //changing index
  IF IrCommand.Next THEN
    IF IrCommand.Index = IrCommandMax THEN
      IrCommand.Index := 0;
      IrCommand.Auto  := false;
    ELSE
      IrCommand.Index := IrCommand.Index + 1;
    END_IF;
    IrCommand.Next    := false;
    IrCommand.Learn   := IrCommand.Auto;
    IrCommand.Test    := false;
    IrCommand.TestOk  := false;
  END_IF;

  IF IrCommand.Prev THEN
    IF IrCommand.Index = 0 THEN
      IrCommand.Index := IrCommandMax;
    ELSE
      IrCommand.Index := IrCommand.Index -1;
    END_IF;
    IrCommand.Prev    := false;
    IrCommand.Learn   := false;
    IrCommand.Test    := false;
    IrCommand.TestOk  := false;
  END_IF;

  //changing names
  IF IrCommand.TextLast <> IrCommand.Text THEN
    IrCommandNames[IrCommand.Index] := IrCommand.Text;
  END_IF;

  IrCommand.Text := IrCommandNames[IrCommand.Index];
  IrCommand.TextLast := IrCommand.Text;

  IF IrCommand.Send THEN
    IrControl.Transmit := true;
    IrControl.CodeTrans := IrCommand.Index+1;
  END_IF;

  //transmiting command
  IF IrCommand.Transmit <> 0 THEN
    IrControl.Transmit := true;
    IrControl.CodeTrans := USINT_TO_UINT(
      DwordBitIndex(In := IrCommand.Transmit)
    );
  END_IF;

  IrControl(Clear :=  IrCommand.Clear,
             ClearAll := IrCommand.ClearAll,
             Learn := IrCommand.Learn,
             CodeLearn := IrCommand.Index+1,
             CodeClear := IrCommand.Index+1,
             IRin := void(C_RI_0401S_IN.IRin),
             IRout := void(C_RI_0401S_OUT.IRout));

//clearing requests
  IF NOT IrControl.Busy THEN
    IrCommand.Transmit := 0;
    IrCommand.Send := 0;
    IrCommand.Clear := 0;
    IrCommand.ClearAll := 0;
  END_IF;

  IrCommand.Learn := IrCommand.Learn AND
                     IrControl.state = sC_RI_0401S_Learning;

  IrCommandLearnEnd(CLK := IrCommand.Learn);

  IrControl.Cancel := IrControl.state = sC_RI_0401S_Learning
                      AND IrCommandLearnEnd.Q;

  IrCommand.Test := IrCommand.Test OR IrControl.LearnDone;

  //identifying of received message
  IF IrControl.ReceiveDone THEN
    IF IrControl.CodeRecv > 0 AND
      IrControl.CodeRecv <= IrCommandMax THEN
      IrCommand.TextRecv := IrCommandNames[IrControl.CodeRecv-1];
    ELSE
      IrCommand.TextRecv := 'unknown';
    END_IF;
    //test result
    IF IrCommand.Test THEN
      IrCommand.TestOk := IrControl.CodeLearn = IrControl.CodeRecv;
      IrCommand.Test   := false;
    END_IF;
  END_IF;

  IrCommand.TestOk := IrCommand.TestOk AND (NOT IrCommand.Learn);

  //automatic learning
  AutoTimeOk(IN := IrCommand.Auto AND IrCommand.TestOk, PT := T#2s);

  IF IrCommand.Auto THEN
    IF AutoTimeOk.Q THEN
      IrCommand.Next := true;
    END_IF;
  END_IF;

END_PROGRAM

6.4  Funkční blok fbRCM2_1
Knihovna : CFoxLib

Funkční blok  fbRCM2_1 je určen pro definici a obsluhu lineárního seznamu ob-
razovek zobrazovaných modulem RCM2-1. Spárování funkčního bloku s konkrétním mo-
dulem je realizováno přes vstupy Press, Counter a RCM_OUT. Do vstupu Press a Coun-
ter je  se  přiřadí  odpovídající  proměnné  ze  vstupní  struktury  modulu  jmenovitě
FLG.PRESS a Counter. Výstupní struktura modulu se předává na vstupu RCM_OUT. Při
volání v jazyku ST je potřeba při předávání této struktury použít funkci  void, jinak bude
hlášen nesoulad datového typu.

Definice obrazovek se provádí pomocí pole struktur předávané na vstupu Screens.
Každá položka pole představuje popis jedné obrazovky. Mezi obrazovkami lze přecházet
pomocí otočného ovladače modulu nebo je možné vnutit  zobrazení obrazovky pomocí
vstupu  SetScr. Při nastavení  SetScr  na TRUE dojde k zobrazení obrazovky s indexem
daným vstupem  ScrNo. Indexy obrazovek začínají jedničkou bez ohledu na indexování
pole definic. Pro přehlednost algoritmu je však vhodné toto pole indexovat od jedné. Pole
může mít libovolný počet položek, který se předá bloku na vstupu MaxScr. MaxScr nesmí
být nikdy větší než rozměr pole!

Výstup Edit indikuje probíhající editaci na aktuálně zobrazené obrazovce. Index ak-
tuálně zobrazené obrazovky udává výstup ActScr.

Definice obrazovky obsahuje zobrazovanou hodnotu (typ REAL nebo TIME), zob-
razené symboly, počet zobrazených desetinných míst a příznaky povolující a indikující
editaci.

Podrobný výčet položek struktury popisujíc obrazovku:

Položka Význam

    RealValue Zobrazovaná hodnota REAL

    RealInc Inkrement při otočení kolečka o jeden krok pro REAL

    RealMin Minimum REAL hodnoty při editaci

    RealMax Maximum REAL hodnoty při editaci

    TimeValue Zobrazovaná hodnota TIME

    TimeMin Minimum TIME hodnoty při editaci

    TimeMax Maximum TIME hodnoty při editaci

    IsTime Příznak pro použití hodnoty TIME místo výchozí REAL

    Editing Probíhá editace

    EditEnd Editace dokončena a potvrzená

    Editable Zobrazenou hodnotu lze editovat pokud je TRUE

    BlinkingDots Zobrazí se blikající dvojtečka

    ShowTenths Zobrazí jedno desetinné místo

    ShowHundredths Zobrazí dvě desetinná místa

    Symbols Symboly které se mají na obrazovce zobrazit. Struktura od-
povídá výstupní struktuře ICO modulu RCM2-1

Popis struktury Symbols

Položka Význam

ONE 
Zobrazí symbol T11

TWO
Zobrazí symbol T22

THREE
Zobrazí symbol T33

FOUR
Zobrazí symbol T44

FIVE
Zobrazí symbol T55

SIX
Zobrazí symbol T66

SEVEN
Zobrazí symbol T77

ERROR Zobrazí symbol P3 ERROR

No Zobrazí symbol P2No.

Clock

Zobrazí symbol T8

Moon

Zobrazí symbol T12

Sun

Zobrazí symbol T13

House

Zobrazí symbol T10

Figure

Zobrazí symbol T11

Drink

Zobrazí symbol T15

Thermometer 

Zobrazí symbol T9

Power

Zobrazí symbol T14  

Setting Položka je ovládána funkčním blokem v režimu editace.
SETTINGS

Celsius
Zobrazí symbol S3°C

Fahrenheit
Zobrazí symbol S4°F

Percent
Zobrazí symbol P6%

rH
Zobrazí symbol W1 rH

dotUp
Nastavení nemá vliv, položka je ovládána funkčním blokem ·

dotDown
Nastavení nemá vliv, položka je ovládána funkčním blokem .

Heating

Zobrazí symbol P7

Cooling

Zobrazí symbol P8

Ventilation

Zobrazí symbol S5

Water

Zobrazí symbol P9

Spanner

Zobrazí symbol P10

P11

Zobrazí symbol P11  

Bell

Zobrazí symbol P12

Manual
Zobrazí symbol W2 M

LN3
Zobrazí symbol P13  

LN2
Zobrazí symbol P14  

LN1
Zobrazí symbol P15  

Rotation

Zobrazí symbol P16

LP1
Zobrazí symbol P17  

LP2
Zobrazí symbol P18  

LP3
Zobrazí symbol P19  

Automatic
Zobrazí symbol P20 A

Minus Nastavení nemá vliv, položka je ovládána funkčním blokem

S2 Nastavení nemá vliv, položka je ovládána funkčním blokem

D1 Nastavení nemá vliv, položka je ovládána funkčním blokem

D2 Nastavení nemá vliv, položka je ovládána funkčním blokem

D3 Nastavení nemá vliv, položka je ovládána funkčním blokem

D4 Nastavení nemá vliv, položka je ovládána funkčním blokem

DE

Zobrazí symbol D5

DOUT Nastaví binární výstup

Rozmístění a označení symbolů a segmentů na displeji RCM2-1

Ovládání displeje:
V běžném režimu lze pomocí tlačítka přecházet mezi nadefinovanými obrazovkami.

Otočení vpravo zobrazí následující obrazovku, otočení vlevo předchozí. Procházení ob-
razovek se děje v kruhu, což znamená, že na poslední obrazovce se otočením po směru
hodinových ručiček zobrazí první obrazovka a na první obrazovce otočením proti směru
se zobrazí poslední obrazovka. Pokud má obrazovka povolenu editaci, stisknutím točítka
dojde k jejímu zahájení.

Editace hodnoty je signalizována blikáním segmentu P1 (SETTINGS). V programu
je během editace nastaven výstup bloku  Edit a položka  Editing v popisu aktuálně zob-
razené  obrazovky  na  TRUE.  Během  editace  není  položka  RealValue (respektive  Ti-
meValue) modifikována dokud nedojde k potvrzení editace zmačknutím točítka. Při po-
tvrzení editace se nastaví na jeden cyklus PLC položka EditEnd. Režim editace je v pří-
padě nečinnosti ukončen automaticky po deseti sekundách s tím, že nedojde ke změně
RealValue (respektive TimeValue) ani k nastavení EditEnd.

Při editaci hodnoty typu REAL bliká počet editovaných cifer daný velikostí inkre-
mentu  RealInc. Otočení o jednu aretaci po směru hodinových ručiček zvýší hodnotu o
RealInc, otočení proti směru ji naopak sníží.

Editace času probíhá ve dvou fázích. Při zahájení editace bliká údaj hodin a oto-
čení o jednu aretaci zvýší nebo sníží  o jednu hodinu. Po potvrzení hodinového údaje
stisknutím točítka se rozblikají a jdou editovat číslice udávající počet minut. Otočením to-
čítka se v této fázi mění údaj po jedné minutě. Další potvrzení stiskem točítka upravenou
hodnotu času uloží

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

Press BOOL Proměnná PRESS ze vstupní struktury 
RCM2-1

Counter SINT Proměnná Counter ze vstupní struktury 
RCM2-1

SetScr BOOL R_EDGE Změní aktuální zobrazovanou obrazovku

ScrNo UINT Číslo obrazovky, na kterou se má přejít při 
nastavení SetScr

MaxScr UINT Počet obrazovek (nesmí být větší než roz-
měr pole definicí obrazovek)

VAR_IN_OUT

Screens TRCM2_1_Screen Struktura modulu TCIB_CRC_BTN

RCM_OUT TLIB_RCM2_OUT Struktura modulu TCIB_CRC_DISP

VAR_OUTPUT

Edit BOOL TRUE znamená, že aktuálně zobrazená 
hodnota je editována

ActScreen UINT Aktuálně zobrazená obrazovka

Následující příklad programu s funkčním blokem fbRCM2_1 definuje 5 obrazovek
umožňující zobrazení aktuálního času, pokojové teploty, editaci dvou žádaných teplot a
nastavení času alarmu. V příkladu je použit časový program TProg2 z knihovny RegoLib.
Pro větší přehlednost je každé z obrazovek v poli definic RCM_Screens definováno jméno
s vlastní inicializací. 

Obrazovka ClockScr zobrazuje aktuální čas, den v týdnu a režim časového progra-
mu.

Obrazovka RoomTempScr zobrazuje teplotu z externího čidla modulu, den v týdnu
a režim časového programu.

Obrazovka ReqTempHiScr umožňuje  editaci žádané teploty TempReqHi.
Obrazovka ReqTempLoScr umožňuje editaci žádané teploty TempReqLo.
Obrazovka AlarmSetScr umožňuje editaci časového údaje TimeAlarm.
Časovač RCM2_1_BackToFirst zajišťuje zobrazení první obrazovky pokud po dobu

jedné minuty nedojde k editaci nebo změně obrazovky.
Pro synchronizaci editovaných hodnot jsou využity funkce RCM2_1_SyncVarReal

a RCM2_1_SyncVarTime.
V příkladu je dále demonstrována možnost dynamického zobrazení symbolů. U ob-

razovek ClockScr a RoomTempScr je indikace stavu časového programu řešena zápisem
do pole definic obrazovek. Zobrazení dne v týdnu je řešeno přímím zápisem do zóny dis-
pleje.

VAR_GLOBAL RETAIN
  WeekProg  : _TimeProg2_Week_ := (
    Mon := ( T_ON1 := T#6h,  T_OFF1 := T#7h,
             T_ON2 := T#16h, T_OFF2 := T#22h),
    Tue := ( T_ON1 := T#6h,  T_OFF1 := T#7h,
             T_ON2 := T#16h, T_OFF2 := T#22h),
    Wed := ( T_ON1 := T#6h,  T_OFF1 := T#7h,
             T_ON2 := T#16h, T_OFF2 := T#22h),
    Thu := ( T_ON1 := T#6h,  T_OFF1 := T#7h,
             T_ON2 := T#16h, T_OFF2 := T#22h),
    Fri := ( T_ON1 := T#6h,  T_OFF1 := T#7h,
             T_ON2 := T#16h, T_OFF2 := T#22h),
    Sat := ( T_ON1 := T#8h,  T_OFF1 := T#22h,
             T_ON2 := T#0s,  T_OFF2 := T#0s),
    Sun := ( T_ON1 := T#8h,  T_OFF1 := T#22h,
             T_ON2 := T#0s,  T_OFF2 := T#0s));
  TempReqLo : REAL := 21.0;
  TempReqHi : REAL := 23.0;
  TimeAlarm : TIME := T#7h;
END_VAR

VAR_GLOBAL
    RCM_Screens : ARRAY [1..5] OF TRCM2_1_Screen;
    ClockScr AT RCM_Screens[1] : TRCM2_1_Screen :=
      (IsTime := true, BlinkingDots := true,
       Symbols := ( Clock := true));
    RoomTempScr AT RCM_Screens[2] : TRCM2_1_Screen :=
      (ShowTenths := true,
       Symbols := ( Thermometer := true, Celsius := true));
    ReqTempHiScr AT RCM_Screens[3] : TRCM2_1_Screen :=
      (RealInc := 0.5,  RealMin := 15.0, RealMax := 30.0,
       ShowTenths := true, Editable := true,
       Symbols := ( Thermometer := true, Sun := true,
        Spanner := true, Celsius := true));
    ReqTempLoScr AT RCM_Screens[4] : TRCM2_1_Screen :=
      (RealInc := 0.5,  RealMin := 15.0, RealMax := 30.0,
       ShowTenths := true, Editable := true,
       Symbols := ( Thermometer := true, Moon := true,
        Spanner := true, Celsius := true));
    AlarmSetScr AT RCM_Screens[5] : TRCM2_1_Screen :=
      (TimeMin := T#0s, TimeMax := T#23h59m,
       IsTime := true,  Editable := true,
       Symbols := ( Clock := true, Spanner := true,
       Bell := true));

END_VAR

PROGRAM prgRCM2_Example
  VAR
    Prog2  : TProg2;
    RCM2_1 : fbRCM2_1;
    RCM2_1_BackToFirst : TON := (PT := T#1m);
    LastScreen : UINT;
  END_VAR

  Prog2(TPg := WeekProg);

  ClockScr.TimeValue := GetTime();
  ClockScr.Symbols.Moon := NOT Prog2.Out;
  ClockScr.Symbols.Sun  := Prog2.Out;
  
  RoomTempScr.RealValue := RCM2_1_IN.eTHERM;
  RoomTempScr.Symbols.Moon := NOT Prog2.Out;
  RoomTempScr.Symbols.Sun  := Prog2.Out;

  RCM2_1_SyncVarReal(Val := TempReqHi, Screen := ReqTempHiScr);

  RCM2_1_SyncVarReal(Val := TempReqLo, Screen := ReqTempLoScr);

  RCM2_1_SyncVarTime(Val := TimeAlarm, Screen := AlarmSetScr);

  RCM2_1(Press   := RCM2_1_IN.FLG.PRESS,
         Counter := RCM2_1_IN.Counter,
         MaxScr := 5,
         Screens := RCM_Screens[1],
         RCM_OUT := void(RCM2_1_OUT));

  IF NOT RCM_Screens[RCM2_1.ActScr].Editable THEN
    RCM2_1_OUT.ICO.ONE   := System_S.COUNTER_DAYS_OF_WEEK = 1;
    RCM2_1_OUT.ICO.TWO   := System_S.COUNTER_DAYS_OF_WEEK = 2;
    RCM2_1_OUT.ICO.THREE := System_S.COUNTER_DAYS_OF_WEEK = 3;
    RCM2_1_OUT.ICO.FOUR  := System_S.COUNTER_DAYS_OF_WEEK = 4;
    RCM2_1_OUT.ICO.FIVE  := System_S.COUNTER_DAYS_OF_WEEK = 5;
    RCM2_1_OUT.ICO.SIX   := System_S.COUNTER_DAYS_OF_WEEK = 6;
    RCM2_1_OUT.ICO.SEVEN := System_S.COUNTER_DAYS_OF_WEEK = 7;
  END_IF;

  RCM2_1_BackToFirst(IN := NOT RCM2_1.Edit AND
                           RCM2_1.ActScr = LastScreen AND
                           RCM2_1.ActScr <> 1);
                     
  LastScreen := RCM2_1.ActScr;

  IF RCM2_1_BackToFirst.Q THEN
    RCM2_1.SetScr := 1;
    RCM2_1.ScrNo  := 1;
  END_IF;

END_PROGRAM

6.5  Funkční blok fbSoilMoisture
Knihovna : CFoxLib

Funkční blok  fbSoilMoisture slouží pro řízení měření objemové vlhkosti půdy pro
moduly C-HM-0308M, C-HM-1113 a C-HM-1121. Blok zajišťuje měření v doporučeném
intervalu 15 minut. Interval měření lze ovlivnit nastavením vstupu T. Probíhající měření je
signalizováno nastavením výstupu Busy. Výsledná hodnota objemové vlhkosti je předává-
na na výstupu Moisture v rozsahu 5 až 50%. V případě přetečení nebo podtečení rozsahu
je výstup OOR nastaven na hodnotu TRUE a  na výstupu Moisture ponechána okrajová
hodnota.

Spojení bloku s příslušným modulem je realizováno přes vstup AI, kde se předává
hodnota analogového vstupu s rozsahem 0 až 2 V a výstupem AO, kde musí být přiřazen
analogový výstup použitý k buzení čidla vlhkosti.

Zapojení a typ čidla viz Příručka projektanta systémů CFox a RFox (TXV 004 16)
kapitola 12.4.2.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

AI REAL Analogový vstup modulu C-HM s rozsahem 0 až 2 V

T TIME Perioda měření

VAR_OUTPUT

Busy BOOL TRUE znamená, že probíhá měření

OOR BOOL TRUE znamená, že znamená že měřená hodnota je 
mimo meze

Moisture REAL Objemová vlhkost půdy 5 až 50%

AO REAL Hodnota pro analogový výstup modulu C-HM

Příklad programu s funkčním blokem fbSoilMoisture:

VAR_GLOBAL
  gSoilMoisture : REAL;
END_VAR

PROGRAM prgSoilMoisture
  VAR
    SoilMoisture : fbSoilMoisture;
  END_VAR

  SoilMoisture(AI := C_HM_0308M_IN.AI.AI1, AO => C_HM_0308M_OUT.AO.AO1);

  gSoilMoisture := SoilMoisture.Moisture;

END_PROGRAM

6.6  Funkční blok fbRead_C_EM_0401M
Knihovna : CFoxLib

Blok fbRead_C_EM_0401M slouží pro čtení dodatečných parametrů z registrů mo-
dulu C-EM-0401M. Čtení je řízeno polem příkazů. 

Pole se definuje jako ARRAY [0..x] OF TC_EM_0401M_Command, kde x je maxi-
mální počet příkazů. Příkazy se do pole plní pomocí následujících funkcí:

C_EM_0401M_Init
C_EM_0401M_ReadLReal
C_EM_0401M_ReadReal
C_EM_0401M_ReadReal2

Funkce pro plnění pole by měly být volány každý cyklus v pevném pořadí. Toto
pořadí udává i pořadí vykonávání příkazů. Změna počtu nebo pořadí příkazů je možná,
měla by však být prováděna jen v případě, že je nastaven výstup done na TRUE, nebo
vstup go na FALSE. V tu chvíli je zajištěno, že blok nemá rozpracován žádný z předcho-
zích příkazů.

Naplněné pole se předává na vstupu C_EM_0401M_Command.

Blok je s hardwarovým modulem svázán přes vstupní a výstupní strukturu AREGI a
AREGO, které se předávají přes stejnojmenné vstupy.

Příkazy se vykonávají cyklicky po dobu, kdy je vstup go nastaven na TRUE.
Po každé, když jsou vykonány všechny příkazy v poli je na jeden cyklus nastaven

výstup done. 
Výstup  actCmd indikuje pořadové číslo příkazu, který aktuálně je vykonáván. Vý-

stup lastCmd číslo příkazu, který byl úspěšně vykonán a výstup errCmd číslo příkazu, je-
hož vykonávání skončilo s chybou.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

Go BOOL aktivace čtení

VAR_OUTPUT

done BOOL všechny příkazy v poli byly vykonány

actCmd UDINT aktuálně zpracovávaný příkaz

lastCmd UDINT poslední zpracovaný příkaz

errCmd UDINT poslední neúspěšně zpracovaný příkaz

VAR_IN_OUT

AREGI TLIB_CEM0401M_ARI Struktura AREGI modulu C-EM-0401M

AREGO TLIB_CEM0401M_ARO Struktura AREGO modulu C-EM-0401M

C_EM_0401M
_Command

TC_EM_0401M_Com-
mand

první člen pole příkazů

Následuje příklad programu s funkčním blokem fbRead_C_EM_0401M. V příkladu
jsou vyčteny napětí na třech fázích (proměnné Uln1 až Uln3), jalové příkony (proměnné
Q1 až Q3) a zdánlivé příkony (proměnné S1 až S3). Navíc jsou každých 10 sekund přidá-
ny příkazy pro čteních celkových spotřeb (proměnné Ep1 až Ep3). 

Přidání příkazů řídí proměnné  readTotalsReq a  readTotals. Proměnná  readTotal-
sReq se nastaví každých 10 sekund. Ve chvíli kdy jsou vykonány všechny příkazy je na
jejím základě nastavena proměnná readTotals, která se shodí opět až ve chvíli, když jsou
vykonány všechny příkazy. Tím je zajištěno, že během vykonávání příkazů se jejich počet
a pořadí nemění.

PROGRAM prgC_EM_0401M_Example
  VAR
    Read_C_EM_0401M : fbRead_C_EM_0401M := ( Go := true);
    C_EM_0401M_Commands : ARRAY [0..10] OF TC_EM_0401M_Command;
    Uln1, Uln2, Uln3, Q1, Q2, Q3, S1, S2, S3 : REAL;
    Ep1, Ep2, Ep3                            : LREAL;
    readTotalsReq, readTotals                : BOOL;
  END_VAR

  C_EM_0401M_Init(size := sizeof(C_EM_0401M_Commands),
                  C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  C_EM_0401M_ReadReal2(reg := 4352,
                       real1 := Uln1, real2 := Uln2,
                       C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  C_EM_0401M_ReadReal (reg := 4356,
                       real1 := Uln3,
                       C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  C_EM_0401M_ReadReal2(reg := 4904,
                       real1 := Q1, real2 := Q2,
                       C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  C_EM_0401M_ReadReal (reg := 4908,
                       real1 := Q3,
                       C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  C_EM_0401M_ReadReal2(reg := 4912,
                       real1 := S1, real2 := S2,
                       C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  C_EM_0401M_ReadReal (reg := 4916,
                       real1 := S3,
                       C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  IF readTotals THEN
    C_EM_0401M_ReadLReal(reg := 8208,
                         lreal1 := Ep1,
                         C_EM_0401M_Command := C_EM_0401M_Commands[0]);
    C_EM_0401M_ReadLReal(reg := 8212,
                         lreal1 := Ep2,
                         C_EM_0401M_Command := C_EM_0401M_Commands[0]);
    C_EM_0401M_ReadLReal(reg := 8216,
                         lreal1 := Ep3,
                         C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  END_IF;

  Read_C_EM_0401M(AREGI := void(C_EM_0401M_IN.AREGI),
                  AREGO := void(C_EM_0401M_OUT.AREGO),
                  C_EM_0401M_Command := C_EM_0401M_Commands[0]);

  IF readTotals AND Read_C_EM_0401M.done THEN
    readTotals := false;
  END_IF;

  readTotalsReq := readTotalsReq OR System_S.R_EDGE_10SEC;

  IF readTotalsReq AND Read_C_EM_0401M.done THEN
    readTotals    := true;
    readTotalsReq := false;
  END_IF;

END_PROGRAM

6.7  Funkční blok fbReadConf_C_EM_0401M
Knihovna : CFoxLib

Blok fbReadConf_C_EM_0401M je variantou bloku fbRead_C_EM_0401M a slouží
ke  čtení dodatečných parametrů z registrů modulu. Navíc dokáže nastavit registry udáva-
jící měřící rozsah proudových traf.

Vstup primaryCT měřící rozsah proudových traf pro měřící vstupy I1 – I3.
Vstup primaryCTN měřící rozsah proudového trafa pro měřící vstup I4.
Měřící rozsahy se nastaví pokud jsou příslušné vstupy  primaryCT a  primaryCTN

nenulové  v  okamžiku,  kdy  se  blok  aktivuje  vstupem  go  nastaveným na  TRUE,  nebo
pokud již je aktivován a dojde ke změně hodnoty jednoho ze vstupů.

V  době  kdy se  parametry nastavují  je  aktivní  výstup  conf a  nezpracovávají  se
uživatelské příkazy.

Popis proměnných :

Proměnná Typ Význam

VAR_INPUT

go BOOL aktivace čtení/konfigurace

primaryCT UINT rozsah měřících proudových traf pro I1-I3
v Ampérech

primaryCTN UINT rozsah měřících proudového trafa pro I4 
v Ampérech

VAR_OUTPUT

done BOOL všechny příkazy v poli byly vykonány

conf BOOL probíhá konfigurace

actCmd UDINT aktuálně zpracovávaný příkaz

lastCmd UDINT poslední zpracovaný příkaz

errCmd UDINT poslední neúspěšně zpracovaný příkaz

VAR_IN_OUT

AREGI TLIB_CEM0401M_ARI Struktura AREGI modulu C-EM-0401M

AREGO TLIB_CEM0401M_ARO Struktura AREGO modulu C-EM-0401M

C_EM_0401M_
Command

TC_EM_0401M_Com-
mand

první člen pole příkazů

Následuje příklad programu s funkčním blokem  fbReadConf_C_EM_0401M. Blok
nejprve nakonfiguruje proudová trafa dle hodnot v inicializaci. Dále jsou v příkladu vyčteny
napětí na třech fázích (proměnné Uln1 až Uln3), jalové příkony (proměnné Q1 až Q3) a
zdánlivé příkony (proměnné S1 až S3). Každých 10 sekund jsou navíc zařazeny příkazy
pro čteních celkových spotřeb (proměnné Ep1 až Ep3). 

Přidání příkazů řídí proměnné  readTotalsReq a  readTotals. Proměnná  readTotal-
sReq se nastaví každých 10 sekund. Ve chvíli kdy jsou vykonány všechny příkazy je na
jejím základě nastavena proměnná readTotals, která se shodí opět až ve chvíli, když jsou
vykonány všechny příkazy. Tím je zajištěno, že během vykonávání příkazů se jejich počet
a pořadí nemění.

PROGRAM prgC_EM_0401M_Example2
  VAR
    Read_C_EM_0401M : fbReadConf_C_EM_0401M := ( go := true,
                                                 primaryCT := 35,
                                                 primaryCTN := 35);
    C_EM_0401M_Commands : ARRAY [0..10] OF TC_EM_0401M_Command;
    Uln1, Uln2, Uln3 : REAL;
    Q1, Q2, Q3       : REAL;
    S1, S2, S3       : REAL;
    Ep1, Ep2, Ep3    : LREAL;
    readTotalsReq    : BOOL;
    readTotals       : BOOL;
  END_VAR

  C_EM_0401M_Init(size := sizeof(C_EM_0401M_Commands),
                  C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  C_EM_0401M_ReadReal2(reg := 4352,
                       real1 := Uln1, real2 := Uln2,
                       C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  C_EM_0401M_ReadReal (reg := 4356,
                       real1 := Uln3,
                       C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  C_EM_0401M_ReadReal2(reg := 4904,
                       real1 := Q1, real2 := Q2,
                       C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  C_EM_0401M_ReadReal (reg := 4908,
                       real1 := Q3,
                       C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  C_EM_0401M_ReadReal2(reg := 4912,
                       real1 := S1, real2 := S2,
                       C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  C_EM_0401M_ReadReal (reg := 4916,
                       real1 := S3,
                       C_EM_0401M_Command := C_EM_0401M_Commands[0]);

  IF readTotals THEN
    C_EM_0401M_ReadLReal(reg := 8208,
                         lreal1 := Ep1,
                         C_EM_0401M_Command := C_EM_0401M_Commands[0]);
    C_EM_0401M_ReadLReal(reg := 8212,
                         lreal1 := Ep2,
                         C_EM_0401M_Command := C_EM_0401M_Commands[0]);
    C_EM_0401M_ReadLReal(reg := 8216,
                         lreal1 := Ep3,
                         C_EM_0401M_Command := C_EM_0401M_Commands[0]);
  END_IF;

  Read_C_EM_0401M(AREGI := void(C_EM_0401M_IN.AREGI),
                  AREGO := void(C_EM_0401M_OUT.AREGO),
                  C_EM_0401M_Command := C_EM_0401M_Commands[0]);

  IF readTotals AND Read_C_EM_0401M.done THEN
    readTotals := false;
  END_IF;

  readTotalsReq := readTotalsReq OR System_S.R_EDGE_10SEC;

  IF readTotalsReq AND Read_C_EM_0401M.done THEN
    readTotals    := true;
    readTotalsReq := false;
  END_IF;

END_PROGRAM

6.8  Funkční blok fbReadConf_C_EM_0300M
Knihovna : CFoxLib

Blok fbReadConf_C_EM_0300M slouží pro konfiguraci proudových traf a čtení do-
datečných parametrů z registrů modulu C-EM-0300M. Čtení je řízeno polem příkazů. 

Pole se definuje jako ARRAY [0..x] OF TC_EM_0300M_Command, kde x je maxi-
mální počet příkazů. Příkazy se do pole plní pomocí následujících funkcí:

C_EM_0300M_Init
C_EM_0300M_ReadLReal
C_EM_0300M_ReadReal
C_EM_0300M_ReadReal2

Funkce pro plnění pole by měly být volány každý cyklus v pevném pořadí. Toto
pořadí udává i pořadí vykonávání příkazů. Změna počtu nebo pořadí příkazů je možná,
měla by však být prováděna jen v případě, že je nastaven výstup done na TRUE, nebo
vstup go na FALSE. V tu chvíli je zajištěno, že blok nemá rozpracován žádný z předcho-
zích příkazů.

Naplněné pole se předává na vstupu C_EM_0300M_Command.

Blok je s hardwarovým modulem svázán přes vstupní a výstupní strukturu AREGI a
AREGO, které se předávají přes stejnojmenné vstupy.

Příkazy se vykonávají cyklicky po dobu, kdy je vstup go nastaven na TRUE.
Po každé, když jsou vykonány všechny příkazy v poli je na jeden cyklus nastaven

výstup done. 
Výstup  actCmd indikuje pořadové číslo příkazu, který aktuálně je vykonáván. Vý-

stup lastCmd číslo příkazu, který byl úspěšně vykonán a výstup errCmd číslo příkazu, je-
hož vykonávání skončilo s chybou.

Vstup primaryCT udává měřící rozsah proudových traf pro měřící vstupy I1 – I3.
Měřící rozsahy se nastaví pokud je vstup primaryCT nenulový v okamžiku, kdy se

blok aktivuje vstupem go nastaveným na TRUE, nebo pokud již je aktivován a dojde ke
změně hodnoty vstupu.

V době  kdy se  parametry nastavují  je  aktivní  výstup  conf a  nezpracovávají  se
uživatelské příkazy.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

go BOOL aktivace čtení/konfigurace

primaryCT UINT rozsah měřících proudových traf pro I1-
I3 v Ampérech

VAR_OUTPUT

done BOOL všechny příkazy v poli byly vykonány

conf BOOL probíhá konfigurace

actCmd UDINT aktuálně zpracovávaný příkaz

lastCmd UDINT poslední zpracovaný příkaz

errCmd UDINT poslední neúspěšně zpracovaný příkaz

VAR_IN_OUT

AREGI TCIB_CEM0300M_ARI Struktura AREGI modulu C-EM-0300M

AREGO TCIB_CEM0300M_ARO Struktura AREGO modulu C-EM-0300M

C_EM_0300M_
Command

TC_EM_0300M_Com-
mand

první člen pole příkazů

Následuje příklad programu s funkčním blokem  fbReadConf_C_EM_0301M. Blok
nejprve nakonfiguruje proudová trafa dle hodnot v inicializaci. Dále jsou v příkladu vyčteny
napětí na třech fázích (proměnné Uln1 až Uln3), jalové příkony (proměnné Q1 až Q3) a
zdánlivé příkony (proměnné S1 až S3). Každých 10 sekund jsou navíc zařazeny příkazy
pro čteních celkových spotřeb (proměnné Ep1 až Ep3). 

Přidání příkazů řídí proměnné  readTotalsReq a  readTotals. Proměnná  readTotal-
sReq se nastaví každých 10 sekund. Ve chvíli kdy jsou vykonány všechny příkazy je na
jejím základě nastavena proměnná readTotals, která se shodí opět až ve chvíli, když jsou
vykonány všechny příkazy. Tím je zajištěno, že během vykonávání příkazů se jejich počet
a pořadí nemění.

PROGRAM prgC_EM_0300M_Example
  VAR
    Read_C_EM_0300M : fbReadConf_C_EM_0300M := ( go := true,
                                                 primaryCT := 30);
    C_EM_0300M_Commands : ARRAY [0..10] OF TC_EM_0300M_Command;
    Uln1, Uln2, Uln3 : REAL;
    Q1, Q2, Q3       : REAL;
    S1, S2, S3       : REAL;
    Ep1, Ep2, Ep3    : LREAL;
    readTotalsReq    : BOOL;
    readTotals       : BOOL;
  END_VAR

  C_EM_0300M_Init(size := sizeof(C_EM_0300M_Commands),
                  C_EM_0300M_Command := C_EM_0300M_Commands[0]);
  C_EM_0300M_ReadReal2(reg := 4352,
                       real1 := Uln1, real2 := Uln2,
                       C_EM_0300M_Command := C_EM_0300M_Commands[0]);
  C_EM_0300M_ReadReal (reg := 4356,
                       real1 := Uln3,
                       C_EM_0300M_Command := C_EM_0300M_Commands[0]);
  C_EM_0300M_ReadReal2(reg := 4904,
                       real1 := Q1, real2 := Q2,
                       C_EM_0300M_Command := C_EM_0300M_Commands[0]);
  C_EM_0300M_ReadReal (reg := 4908,
                       real1 := Q3,
                       C_EM_0300M_Command := C_EM_0300M_Commands[0]);
  C_EM_0300M_ReadReal2(reg := 4912,
                       real1 := S1, real2 := S2,
                       C_EM_0300M_Command := C_EM_0300M_Commands[0]);
  C_EM_0300M_ReadReal (reg := 4916,
                       real1 := S3,
                       C_EM_0300M_Command := C_EM_0300M_Commands[0]);

  IF readTotals THEN
    C_EM_0300M_ReadLReal(reg := 8208,
                         lreal1 := Ep1,
                         C_EM_0300M_Command := C_EM_0300M_Commands[0]);
    C_EM_0300M_ReadLReal(reg := 8212,
                         lreal1 := Ep2,
                         C_EM_0300M_Command := C_EM_0300M_Commands[0]);
    C_EM_0300M_ReadLReal(reg := 8216,
                         lreal1 := Ep3,
                         C_EM_0300M_Command := C_EM_0300M_Commands[0]);
  END_IF;

  Read_C_EM_0300M(AREGI := void(C_EM_0300M_IN.AREGI),
                  AREGO := void(C_EM_0300M_OUT.AREGO),
                  C_EM_0300M_Command := C_EM_0300M_Commands[0]);

  IF readTotals AND Read_C_EM_0300M.done THEN
    readTotals := false;
  END_IF;

  readTotalsReq := readTotalsReq OR System_S.R_EDGE_10SEC;

  IF readTotalsReq AND Read_C_EM_0300M.done THEN
    readTotals    := true;
    readTotalsReq := false;
  END_IF;

END_PROGRAM

Výrobce si vyhrazuje právo na změny dokumentace. Poslední aktuální vydání je k dispo-
zici na internetu www.tecomat.cz

